AD NUMBER

AD913117

LIMITATION CHANGES

TO:

Approved for public release; distribution is unlimited.

FROM:

Distribution authorized to U.S. Gov't. agencies only; Test and Evaluation; 17 MAR 1971. Other requests shall be referred to Electronic System Div., Hanscom AFB, MA.

AUTHORITY

ESD ltr 4 Dec 1973

Project Report

PA-229-1
(RSP)

Data Reduction Program Documentation
ALTAIR Tape Read Package

(Effective: April 1970)

C. R. Berndtson
R. H. French
D. E. Nessman

19595

17 March 1971

Prepared for the Advanced Research Projects Agency, the Department of the Army, and the Department of the Air Force under Electronic Systems Division Contract F19628-70-C-0230 by

Lincoln Laboratory
MASSACHUSETTS INSTITUTE OF TECHNOLOGY
Lexington, Massachusetts



Unclassified

MASSACHUSETTS INSTITUTE OF TECHNOLOGY

LINCOLN LABORATORY

(6)

DATA REDUCTION PROGRAM DOCUMENTATION
ALTAIR TAPE READ PACKAGE
(EFFECTIVE: APRIL 1970)

(10) Charles

C. R. BERNDTSON,
~~Group 92~~

R. H. FRENCH
D. E. NESSMAN

Philco - Ford Corporation
Editors

(12) 26p.

(14) PA-229-1

(15) F19628-70-C-0230,
ARPA Order-600

(9) PROJECT REPORT, PA-229-1 (RSP)

(11) 17 MAR[redacted]71

L. G. Hanscom Field
Bedford, Mass. 01730

LEXINGTON MASSACHUSETTS

Unclassified

207 650

The work reported in this document was performed at Lincoln Laboratory, a center for research operated by Massachusetts Institute of Technology. This work was sponsored by the Advanced Research Projects Agency of the Department of Defense (ARPA Order 600), the Department of the Army, and the Department of the Air Force under Air Force Contract F19628-70-C-0230.

FOREWORD

This is the first report in the Data Reduction Program Documentation series. All reports will bear the document number PA-229 and will be distinguished by post-numerals -- the present report being PA-229-1.

The series will document in detail the programs used at Lincoln Laboratory in preparing the PRESS Operation and Data Summary (POD) Reports. Each POD is a presentation of data from a ballistic missile flight into the vicinity of the Kiernan Reentry Measurements Site (KREMS), situated on the island of Roi-Namur in the Kwajalein Atoll of the Marshall Islands. The data collected at KREMS is ~~shipped to Lincoln and there~~ validated, edited, and processed, A selected subset of the data -- perhaps one-tenth to one percent of the whole -- is transcribed to computer formatted magnetic tapes and processed through a library of programmed algorithms to produce the information displayed, either graphically or in tabular form, in PODs, and subsequently used for interpretation and analysis.

It should be noted that, although KREMS possesses both radar and optical instrumentation, the emphasis in this report series is on the treatment of radar data. It should also be noted -- and in fact is to be stressed -- that an appreciation of the details of an algorithm must be based on an understanding

of the characteristics and idiosyncrasies of the radar which generated the data. It is not the purpose of this series to describe the ALTAIR, ALCOR, or TRADEX radars: descriptions are found in other references, either already published or in preparation. The reader is urged to familiarize himself with the instrument characteristics.

Within this series, each program will be documented as it is written for the Lincoln IBM 360/67 computer, and it is planned that each major program will constitute a separate report. To a certain extent, each report will be self-contained: abbreviations, symbols, minor subroutines, etc., will be repeated in each report even if common to a number of programs. However, a few subroutines are voluminous enough to warrant separate reports, and programs using these subroutines will cross-reference the appropriate report.

Although many of the programs produce, as output, graphical displays on a Stromberg-Carlson or Calcomp plotter, no attempt is made to document the control instructions for such devices, since the instructions are standardized and available elsewhere.

Each report will be dated according to the date of completion of the documentation. No implication is made that a program will not subsequently be modified, amended, or superseded: on the contrary, the history of radar data processing is one of continuous evolution of techniques, and it is unrealistic to assume that steady-state has been reached. The PA-229 series is being published for the convenience of interested parties, and Lincoln assumes no responsibility for the correctness of the information presented, nor for its currency.

The preparation of reports in this series is under the Editorship of Charles R. Berndtson of Lincoln, and of D. Nessmann and R. French of Philco-Ford Corporation. Inquiries, suggestions, corrections, criticisms, and requests for additional copies should be directed to C. R. Berndtson.

Finally, it will be observed that, in some cases, a report recognizes by name a principal contributor; in other cases, no name is given. Due to the intricate, evolutionary manner in which the programs came into being, the editors regret that it is in general impossible to give due credit to all -- mathematicians or radar analysts or programmers -- who contributed to the definition and writing of the programs.

The work reported in this document was performed at Lincoln Laboratory, a center for research operated by Massachusetts Institute of Technology. This work was sponsored by the Advanced Research Projects Agency of the Department of Defense (ARPA Order 600), the Department of the Army, and the Department of the Air Force under Air Force Contract F19628-70-C-0230.

Alan A. Grometstein

AAG:jb

CONTENTS

ALTAIR TAPE READ PACKAGE

The ALTAIR Tape Read Package retrieves data from the catalog and transcription tapes. It contains four IBM 360 assembler language subroutines which are called by the user's Fortran program.

Subroutine	Entry
BREADS	BREADS BREAD REW
HDRR	HDRR NAMET
FORM	FORM
GETS	GET IGET

Appendices A-D present program listings of these subroutines.

The user's program calls a Fortran subroutine (CHEAD for the catalog tape and THEAD for the transcription tape) which in turn calls the ALTAIR Tape Read Package through the above entries. Entry BREADS is called to define the buffer areas and various flag and parameter locations used in the reading process. BREADS also opens the file and causes the first record of an ALTAIR tape to be read. The data from the first record is not available to the user, however, until the reading call BREAD is given. The BREAD call moves the data from the first record into an area where it is available to the Fortran user and also initiates the reading of the second record.

Entry HDRR is called to gain access to Tape Header Record FMHDRD which is the first record of a catalog or transcription tape. The remaining format and calibration records at the front of the tape are now read by successive BREAD calls.

Each of these records is interrogated by Entry NAMET which checks to see if the record should be stored in core. Entry FORM is called to store and unpack selected calibration and format records in a convenient form for later use by the GETS subroutine. Once a data record is reached, control will return to the main processing program from CHEAD or THEAD.

As successive data records are read into memory by the BREAD call, particular data items may be acquired by means of the entries GET (floating point number) and IGET (binary integer). A detailed description of the use and function of these subroutines is provided at the front of each subroutine listing.

COMMON SYMBOLS AND ABBREVIATIONS

(The units given for certain quantities are the units commonly used for those quantities, unless otherwise noted.)

ADT	ALCOR Data Tape
Alt	Altitude (km)
APS	Average Pulse Shape
ARS	ALTAIR Recording System
Avg	Average, Averaging
Az	Azimuth (deg)
CADJ	Adjusted Calibration Constant (db)
C-band	ALCOR frequency, 5664 MHz (NB) and 5667 MHz (WB)
El	Elevation (deg)
EOF	End of File
GMT	Greenwich Mean Time
h	Hours
Hz	Hertz
in	Inches
LC	Left Circular Polarization
min	Minutes
NB	Narrow Band
NRTPOD	Non-real Time Precision Orbit Determination Program
POD	Project PRESS Operation and Data Summary Report
Phase	Presented in deg
PRF	Pulse Repetition Frequency (pps)
PRI	Pulse Repetition Interval (s)
pps	Pulses per second
pts	Points

R	Range (km)
$\dot{R}$	Range Rate (km/s)
rad	Radians
RC	Right Circular Polarization
RCS	Radar Cross Section (dbsm)
s	Seconds
SD_w	Standard Deviation of Wake Velocity
T	Time
TAL	Time After Launch (s)
UHF	ALTAIR Frequency; 415 MHz
V	Velocity
V_d	Doppler Velocity
V_w	Mean Wake Velocity
VHF	ALTAIR Frequency; 155.5 MHz
WB	Wide Band
θ	Total Off-axis Angle (deg)
λ	Wavelength
*	Denotes Multiplication

APPENDIX A
SUBROUTINE BREADS PROGRAM LISTING

```
***EREADS              DOUBLE BUFFERED READ FOR FORTRAN                  BRE00010
*                                                                        BRE00020
*       THIS SUBROUTINE PROVIDES A DOUBLE BUFFERED READ CAPABILITY FOR   BRE00030
* FORTRAN READING OF UNFORMATED BINARY TAPES.                            BRE00040
*                                                                        BRE00050
*       TO UTILIZE THIS ROUTINE TWO CALLS, BREADS AND BREAL, ARE         BRE00060
* NECESSARY. EREADS DEFINES THE BUFFER AREAS AND VARIOUS FLAG AND        BRE00070
* PARAMETER LOCATIONS USED IN THE READING PROCESS. BREADS ALSO OPENS     BRE00080
* THE FILE AND CAUSES THE FIRST RECORD OF AN ALTAIR TAPE TO BE READ.     BRE00090
* THE DATA FROM THE FIRST RECORD IS NOT AVAILABLE TO THE USER HOWEVER,   BRE00100
* UNTIL THE READING CALL BREAD IS GIVEN. THE BREAD CALL MOVES THE DATA   BRE00110
* FROM THE FIRST RECORD INTO AN AREA WHERE IT IS MADE AVAILABLE TO THE   BRE00120
* FORTRAN USER AND ALSO INITIATES THE READING OF THE SECOND RECORD.      BRE00130
* SUCCEEDING RECORDS ARE NOW READ BY THE BREAD CALL.                     BRE00140
*                                                                        BRE00150
* CAILING SEQUENCES                                                      BRE00160
*                                                                        BRE00170
* CALL BREADS(LN,IBUF1,IBUF2,MAX,IFL,INDX,LEN,IFLG,IADD)                 BRE00180
*                                                                        BRE00190
* WHERE  LN     IS NOT APPLICABLE TO OS. DUMMY PARAMETER.                BRE00200
*        IBUF1  IS THE BEGINNING OF THE FIRST BUFFER.                    BRE00210
*        IBUF2  IS THE BEGINNING OF THE SECOND BUFFER.                   BRE00220
*        MAX    IS THE MAXIMUM RECORD LENGTH TO BE READ.                 BRE00230
*        IFL    IS A FILE COUNTER LOCATION WHICH IS INCREMENTED BY ONE   BRE00240
*               FOR EACH END OF FILE MARK DETECTED.                      BRE00250
*        INDX   IS SET TO EITHER 1 OR 2 AND DEFINES WHICH BUFFER WAS     BRE00260
*               USED TO READ A RECORD.                                   BRE00270
*        LEN    IS SET TO GIVE THE ACTUAL RECORD LENGTH READ.            BRE00280
*        IFLG   IS A FLAG LOCATION USED TO STORE PARITY AND END OF FILE  BRE00290
*               INDICATORS. 1 = NORMAL READ, 2 = PARITY ERROR, AND       BRE00300
*               3 = EOF.                                                 BRE00310
*        IADD   IS SET TO THE ACTUAL ADDRESS OF THE READ BUFFER USED FOR BRE00320
*               READING.                                                 BRE00330
*                                                                        BRE00340
* CAIL BREAD(LN)                                                         BRE00350
*                                                                        BRE00360
*       THIS COMPLETES THE READING OF A RECORD FROM THE SPECIFIED TAPE   BRE00370
* AND INITIATES THE READ OF THE NEXT RECORD. THIS CALL ALSO SETS THE     BRE00380
* THE FILE COUNT, INDEX VALUE, LENGTH, FLAG AND BUFFER ADDRESS TO THE    BRE00390
* PROPER VALUES FOR THE READ JUST COMPLETED.                             BRE00400
*                                                                        BRE00410
* CAIL REW(LN)                                                           BRE00420
*                                                                        BRE00430
*       REWINDS THE TAPE.                                                BRE00440
*                                                                        BRE00450
*       VERSION 02/19/70                                                 BRE00460
*                                                                        BRE00470
        MACRO                                                           BRE00480
&N      SSAVA &GR,&SA                                                   BRE00490
&N      BALR  &GR,0                                                     BRE00500
        USING *,&GR                                                     BRE00510
        LR    12,13                                                     BRE00520
        LA    13,&SA                                                    BRE00530
        ST    12,4(0,13)                                                BRE00540
        ST    13,8(0,12)                                                BRE00550
```

```
         DROP  15                                                  BRE00560
         MEND                                                      BRE00570
         ENTRY BREAD                                               BRE00580
         ENTRY REW                                                 BRE00590
         ENTRY BREADS                                              BRE00600
         USING *,15                                                BRE00610
BREADS   STM   14,12,12(13)                                        BRE00620
         LR    12,13                                               BRE00630
         LA    13,SAVE                                             BRE00640
         ST    12,4(0,13)                                          BRE00650
         ST    13,8(0,12)                                          BRE00660
         ST    1,SAVEP                                             BRE00670
         MVI   CC,X'00'                                            BRE00680
         MVI   TAPCE,X'FF'                                         BRE00690
         LM    2,10,0(1)                                           BRE00700
         L     2,0(0,2)                                            BRE00710
         ST    2,UNIT                                              BRE00720
         ST    3,ABUF1                                             BRE00730
         ST    3,CURBF                                             BRE00740
         ST    4,ABUF2                                             BRE00750
         L     5,0(0,5)                                            BRE00760
         ST    5,LENGTH                                            BRE00770
         ST    6,IFL                                               BRE00780
         SR    5,5                                                 BRE00790
         ST    5,0(0,6)                                            BRE00800
         ST    7,INDX                                              BRE00810
         ST    8,LEN                                               BRE00820
         ST    9,IFLG                                              BRE00830
         N     10,CON1                                             BRE00840
         ST    10,IADD                                             BRE00850
         L     8,CURBF                                             BRE00860
         USING IHADCB,5                                            BRE00870
         LA    5,RDCB                                              BRE00880
         TM    DCBOFLGS,X'10'                                      BRE00890
         BNO   BREAD2                                              BRE00900
         MVI   SHOW,X'70'                                          BRE00910
         B     BREAD1                                              BRE00920
BREAD2   LR    9,15                                                BRE00930
         CLOSE (RDCB,REREAD)                                       BRE00940
         LR    15,9                                                BRE00950
         OPEN  RDCB                                                BRE00960
         LR    15,9                                                BRE00970
         OI    DCBIFLGS,X'0C'                                      BRE00980
BREAD3   MVI   KERR,X'00'                                          BRE00990
         MVI   SHOW,X'01'                                          BRE01000
         MVI   CC,X'FF'                                            BRE01010
         B     BREAD1                                              BRE01020
         USING *,15                                                BRE01030
REW.     STM   14,12,12(13)                                        BRE01040
         LR    12,13                                               BRE01050
         LA    13,SAVE                                             BRE01060
         ST    12,4(0,13)                                          BRE01070
         ST    13,8(0,12)                                          BRE01080
         LR    9,15                                                BRE01090
         CLOSE (RDCB,REREAD)                                       BRE01100
```

```
         LR    15,9                                                  BRE01110
         MVI   IFL+3,X'00'                                           BRE01120
         MVI   EOFCTR+3,X'01'                                        BRE01130
         MVI   CC,X'00'                                              BRE01140
         L     13,SAVE+4                                             BRE01150
         RETURN (14,12),RC=0                                         BRE01160
         USING *,15                                                  BRE01170
BREAD    STM   14,12,12(13)                                          BRE01180
          LR   12,13                                                 BRE01190
         LA    13,SAVE                                               BRE01200
         ST    12,4(0,13)                                            BRE01210
         ST    13,8(0,12)                                            BRE01220
BREAD1   BALR  11,0                                                 BRE01230
         USING *,11                                                  BRE01240
FILBUF   L     8,CURBF                                               BRE01250
         LA    7,5                                                   BRE01260
         ST    7,CTR                                                 BRE01270
         TM    SHOW,X'70'                                            BRE01280
         BO    CK                                                    BRE01290
          TM   SHOW,X'01'                                            BRE01300
         BO    RDAHD                                                 BRE01310
          TM   CC,X'FF'                                              BRE01320
         BC    CK                                                    BRE01330
RD       MVI   KERR,X'00'                                            BRE01340
         READ  BRDCB,SF,RDCB,ABUFA,X'2000'                           BRE01350
         MVI   SHOW,X'02'                                            BRE01360
         TM    CC,X'FF'                                              BRE01370
         BO    RETURN                                                BRE01380
         TM    TAPCE,X'FF'                                           BRE01390
         BNO   RDAHD                                                 BRE01400
CK       CHECK BRDCB                                                 BRE01410
         TM    SHOW,X'70'                                            BRE01420
         BNO   NEXT                                                  BRE01430
BBSP     BSP   RDCB                                                  BRE01440
         MVI   KERR,X'00'                                            BRE01450
          MVI  SHOW,X'01'                                            BRE01460
         MVI   CC,X'FF'                                              BRE01470
         B     FILBUF                                                BRE01480
NEXT     MVI   SHOW,X'03'                                            BRE01490
         TM    KERR,X'FF'                                            BRE01500
         BO    ERROR                                                 BRE01510
         BAL   14,MOVE                                               BRE01520
RDAHD    LA    8,1                                                   BRE01530
         L     7,IFLG                                                BRE01540
         ST    8,0(0,7)                                              BRE01550
         SR    7,7                                                   BRE01560
         L     8,CURBF                                               BRE01570
         LH    7,2(0,8)                                              BRE01580
         L     8,LEN                                                 BRE01590
         ST    7,0(0,8)                                              BRE01600
         BAL   5,SWITCH                                              BRE01610
         MVI   TAPCE,X'00'                                           BRE01620
         MVI   CC,X'FF'                                              BRE01630
         L     8,CURBF                                               BRE01640
         LA    7,5                                                   BRE01650
```

```
         ST    7,CTR                                                    BRE01660
         B     RD                                                       BRE01670
RETURN   L     13,SAVE+4                                                BRE01680
         MVI   SHOW,X'04'                                               BRE01690
         LM    14,12,12(13)                                             BRE01700
         RETURN RC=0                                                    BRE01710
SWITCH   L     8,CURBF                                                  BRE01720
         L     6,INDX                                                   BRE01730
         L     7,IADD                                                   BRE01740
         ST    8,0(0,7)                                                 BRE01750
         S     8,ABUF1                                                  BRE01760
         BZ    BB                                                       BRE01770
         L     8,ABUF1                                                  BRE01780
         LA    9,2                                                      BRE01790
         B     SSWA                                                     BRE01800
BB       L     8,ABUF2                                                  BRE01810
         LA    9,1                                                      BRE01820
SSWA     ST    8,CURBF                                                  BRE01830
         ST    9,0(0,6)                                                 BRE01840
         BR    5                                                        BRE01850
RDCB     DCB   DSORG=PS,MACRF=RC,DEVD=TA,DEN=2,TRTCH=C,RECFM=U,         XBRE01860
               BLKSIZE=8132,BUFNO=2,EODAD=EOF,SYNAD=TPSYN,DDNAME=RALT7,XBRE01870
               BFTEK=S                                                  BRE01880
TPSYN    MVI   KERR,X'FF'                                               BRE01890
         BR    14                                                       BRE01900
EOF      LA    8,3                                                      BRE01910
         L     7,IFLG                                                   BRE01920
         ST    8,0(0,7)                                                 BRE01930
         L     8,EOFCTR                                                 BRE01940
         A     8,ONE                                                    BRE01950
         ST    8,EOFCTR                                                 BRE01960
         L     7,IFL                                                    BRE01970
         ST    8,0(0,7)                                                 BRE01980
         MVI   TAPCE,X'00'                                              BRE01990
         MVI   CC,X'FF'                                                 BRE02000
         B     RD                                                       BRE02010
ERROR    L     7,CTR                                                    BRE02020
         BCT   7,*+4                                                    BRE02030
         ST    7,CTR                                                    BRE02040
         MVI   CC,X'00'                                                 BRE02050
         MVI   TAPCE,X'FF'                                              BRE02060
         LTR   7,7                                                      BRE02070
         BNZ   BSP                                                      BRE02080
         MVI   TAPCE,X'00'                                              BRE02090
         LA    7,2                                                      BRE02100
         L     8,IFLG                                                   BRE02110
         ST    7,0(0,8)                                                 BRE02120
         MVI   CC,X'FF'                                                 BRE02130
         B     RD                                                       BRE02140
BSP      BSP   RDCB                                                     BRE02150
         B     RD                                                       BRE02160
MOVE     L     2,CURBF                                                  BRE02170
         L     9,LENGTH                                                 BRE02180
         AR    9,2                                                      BRE02190
         LA    8,4                                                      BRE02200
```

```
         LA     3,ABUFA                                          BRE02210
LOOP  MVC       0(224,2),0(3)                                    BRE02220
      MVC       224(224,2),224(3)                                BRE02230
         MVC    448(224,2),448(3)                                BRE02240
         MVC    672(224,2),672(3)                                BRE02250
         MVC    896(224,2),896(3)                                BRE02260
         MVC    1120(220,2),1120(3)                              BRE02270
         MVC    1340(220,2),1340(3)                              BRE02280
         MVC    1560(220,2),1560(3)                              BRE02290
         MVC    1780(220,2),1780(3)                              BRE02300
         MVC    2000(48,2),2000(3)                               BRE02310
         A      3,T2048                                          BRE02320
         A      2,T2048                                          BRE02330
         CR     9,2                                              BRE02340
         BCR    13,14                                            BRE02350
         BCT    8,LOOP                                           BRE02360
         BR     14                                               BRE02370
T2048    DC     F'2048'                                          BRE02380
CON1     DC     X'000FFFFF'                                      BRE02390
KEBR     DC     X'00'                                            BRE02400
TAPCE    DC     X'00'                                            BRE02410
CC       DC     X'00'                                            BRE02420
SHOW     DC     X'0F'                                            BRE02430
EOFCTR   DC     F'1'                                             BRE02440
CTR      DC     F'0'                                             BRE02450
UNIT     DC     F'0'                                             BRE02460
SAVEP    DC     F'0'                                             BRE02470
CURBF    DC     F'0'                                             BRE02480
ABUF1    DC     F'0'                                             BRE02490
ABUF2    DC     F'0'                                             BRE02500
LENGTH   DC     F'0'                                             BRE02510
IFL      DC     F'0'                                             BRE02520
INDX     DC     F'0'                                             BRE02530
LEN      DC     F'0'                                             BRE02540
IFLG     DC     F'0'                                             BRE02550
IADD     DC     F'0'                                             BRE02560
SAVE     DC     18F'-1'                                          BRE02570
ONE      DC     F'1'                                             BRE02580
ABUFA    DS     2048F                                            BRE02590
         DCBD                                                    BRE02600
         END                                                     BRE02610
```

APPENDIX B
SUBROUTINE HDRR PROGRAM LISTING

HDRR — RETRIEVES DATA FROM TAPE HEADER RECORD

THIS SUBROUTINE HAS TWO ENTRIES, HDRR AND NAMET. HDRR IS USED TO GAIN ACCESS TO TAPE HEADER RECORD FMHDRD WHICH APPEARS AS THE FIRST RECORD OF A RADAR, CATALOG, OR TRANSCRIPTION TAPE. NAMET IS USED TO CONVERT THE 4 CHARACTER BCD NAMES OF OTHER DATA BLOCK FORMATS TO EBCDIC. THIS PERMITS THE FORTRAN PROGRAM TO BYPASS THE STORAGE OF FORMAT TABLES NOT NEEDED BY THE PROGRAM.

```
* CALLING SEQUENCES
*
* CALL HDRR(IADD,ITYP,IDARR)
*
* WHERE  IADD   IS SET TO THE ACTUAL ADDRESS OF THE READ BUFFER USED
*               FOR READING.
*        ITYP   INDICATES THE TYPE OF TAPE; INTEGER VARIABLE.
*               WHERE:
*                     1 = CATALOG TAPE
*                     2 = TRANSCRIPTION TAPE
*                     5 = RADAR TAPE
*        IDARR  DATA TAPE IDENTIFICATION IN EBCDIC. DIMENSIONED EIGHT
*               WORDS. CONTAINS CLASSIFIED INFORMATION SO SHOULD ONLY
*               BE USED WITH DISCRETION.
*
* CALL NAMET(IADD,NAMED)
*
* WHERE  IADD   IS SET TO THE ACTUAL ADDRESS OF THE READ BUFFER USED
*               FOR READING.
*        NAMED  IS THE 4 CHARACTER EBCDIC NAME OF THE FORMAT TABLE
*               MOST RECENTLY PROCESSED BY THE READ ROUTINE.
*
*        VERSION 11/20/69
*
         START  0
         ENTRY  NAMET
         ENTRY  HDRR
         USING  *,15
HDRR     STM    14,12,12(13)
         LR     12,13
         LA     13,SAVE
         ST     12,4(0,13)
         ST     13,8(0,12)
         LM     2,4,0(1)
         L      2,0(0,2)
         L      6,8(0,2)
         ST     6,0(0,3)
         LA     3,3
         LR     5,4
         LA     9,SAVER
LOOPTW   LM     6,8,12(2)
         STM    6,8,0(9)
         SRDL   6,2
         SIL    6,2
         SLDL   6,6
         SLL    6,2
```

```
          SIDL  6,6
          SIL   6,2
          SLDL  6,6
          ST    6,0(0,4)
          LR    7,8
          SR    6,6
          SLL   7,8
          SLDL  6,6
          SLL   6,2
          SLDL  6,6
          SLL   6,2
          SLDL  6,6
          SLL   6,2
          SLDL  6,6
          ST    6,4(0,4)
          A     4,=F'8'
          A     2,=F'12'
          A     9,=F'12'
          BCT   3,LOOPTW
TRNSLVN   TR    0(24,5),TABLE
          L     13,SAVE+4
          LM    14,12,12(13)
          RETURN RC=0
      USING     *,15
NAMET     STM   14,12,12(13)
          LR    12,13
          EALR  11,0
          USING *,11
          LA    13,SAVE
          ST    12,4(0,13)
          ST    13,8(0,12)
          LM    4,5,0(1)
          L     4,0(0,4)
          LM    2,3,0(4)
          N     2,=X'000000FF'
          SRDL  2,2
          SLL   2,2
          SLDL  2,6
          SLL   2,2
          SLDL  2,6
          SLL   2,2
          SLDL  2,6
          ST    2,0(0,5)
          TR    0(4,5),TABLE
          L     13,SAVE+4
          LM    14,12,12(13)
          RETURN RC=0
SAVE      DC    18F'-1'
SAVEF     DC    20F'15'
TABLE     DC    C'0123456789 ='
          DC    C'''    /STUVWX'
          DC    C'YZ .)   -JKL'
          DC    C'MNOPQR $*   '
          DC    C'+ABCDEFGHI ,('
          END
```

APPENDIX C

SUBROUTINE FORM PROGRAM LISTING

```
***FORM                    STORES DATA BLOCK FORMATS
*
*       THIS SUBROUTINE STORES AND UNPACKS SELECTED DATA BLOCK FORMATS
* IN A CONVENIENT FORM FOF LATER USE BY THE GET/IGET FUNCTIONS.
*
* CALLING SEQUENCES
*
* CALL FORM(IADD,ITEM,IB,NAMED,NTEM,ARG6)
*
* WHERE  IADD   IS SET TO THE ACTUAL ADDRESS OF THE READ BUFFER USED
*               FOR READING.
*        ITEM   IS THE FORMAT TAPE IMAGE. EACH ITEM IS STORED IN SIX
*               HALFWORDS. ITEM MUST BE DIMENSIONED TO THE TOTAL NUMBER
*               OF ITEMS STORED X 6.
*        IB     IS THE ITEM BASE INDEX.
*        NAMED  IS THE 4 CHARACTER EBCDIC NAME OF THE FORMAT TABLE
*               MOST RECENTLY PROCESSED BY THE READ ROUTINE.
*        NTEM   IS THE NUMBER OF ITEMS IN THE MOST RECENTLY READ FORMAT
*               TABLE.
*        ARG6   IS AN ABNORMAL RETURN; &STATEMENT LABEL. OCCURS WHEN A
*               FORMAT TABLE HAS MORE THAN 700 ITEMS OR WHEN A FORMAT
*               TABLE HAS 0 LENGTH.
*
*        VERSION 10/27/69
*
         START 0
         ENTRY FORM                    SEARCH FORMAT TABLES
         USING *,15
FORM     B     F5
         DC    XL1'05',CL5'FORM'       ENTRY
F5       STM   2,10,28(13)             SAVE FORTRAN REGISTERS USED
         LM    5,9,0(1)                LOAD INDEXES TO TABLES
*                                      5  IADD  - TAPE I/O AREA
*                                      6  ITEM  - FORMAT TAPE IMAGE
*                                      7  IB    - ITEM BASE INDEXES
*                                      8  NAME  - FORMAT NAMES
*                                      9  NTEM  - NUMBER OF ITEMS/FORM
         ST    6,0(7)                  BASE ADDRESS IN BYTES
         L     5,0(5)                  LOAD ADDRESS
         MVC   WD+1(3),0(5)            STANDARD WORD BOUNDRY
         NI    WD,X'00'
         LH    4,WD+2                  LOAD 2 BYTE LENGTH OF FORMAT
         SRL   4,1                     DIVIDE FORMAT LENGTH BY 2
         S     4,=F'1'                 NUMBER OF ITEMS
         LR    10,4                    SAVE NO. OF ITEMS +1
         ST    4,0(9)                  STORE NTEM
         MVC   WD+1(3),3(5)            24 BIT NAME
         L     2,WD                    LOAD BCD NAME
         LA    1,4                     4 BCD LETTERS
F10      SRDL  2,6                     6 BCD BITS/BYTE
         SRL   3,2                     2 ZERO BITS/BYTE
         BCT   1,F10
         ST    3,0(8)                  STORE NAME IN BCD
         TR    0(4,8),TABLE            TRANSLATE NAME INTO EBCDIC
         C     4,=F'0'                 TEST FOR ZERO TABLE LENGTH
```

```
         BNH   F45
         C     4,=F'700'   10/27/69      TEST FOR TOO LONG
         BNL   F45
F15      LA    5,6(5)                    NEXT ITEM
         MVC   WD+1(3),0(5)
         L     4,WD
         N     4,=X'007FFFFF'            DROP 8 BIT
         SR    2,2
         LR    3,4
         D     2,=F'100000'
         STH   3,0(6)                    P
         M     2,=F'100000'
         SR    4,3
         SR    2,2
         LR    3,4
         D     2,=F'10000'
         STH   3,2(6)                    M
         M     2,=F'10000'
         SR    4,3
         STH   4,4(6)                    RRRR
         SR    2,2
         MVC   WD+1(3),3(5)
         CH    2,2(6)
         BNE   F35                       MODE ZERO
         MVC   8(4,6),2(5)               360 FLOATING
         CH    2,0(6)                    BYTE ADDRESS
         BNE   F40
         L     7,WD                      MANTISSA
         SLL   7,8
         SR    9,9
         CR    7,2                       NEGATIVE
         BNL   F20
         LA    9,1
         SLL   7,1                       DROP SIGN BIT
         SRL   7,1
F20      SRL   7,8
         TM    0(5),X'80'
         BO    F25
         TM    3(5),X'40'
         BO    F25
         O     7,=X'46000000'            INTEGER EXPONENT
         CH    2,4(6)
         BE    F30
F25      SLL   7,8                       DROP EXPONENT
         SRL   7,8
         IH    2,4(6)
         SLL   2,23
         SRA   2,23
         AH    2,=X'0103'
         SRDL  2,2
         SRL   3,30
         SLL   7,0(3)
         SRL   7,2
         SRDL  2,8
         OR    7,3
```

```
F30       SLL    9,31
          OR     7,9
          ST     7,8(6)
          B      F40
F35       L      4,WD
          LR     3,4
          D      2,=F'10000'
          STH    3,6(6)                    XX
          M      2,=F'10000'
          SR     4,3
          SR     2,2
          LR     3,4
          D      2,=F'100'
          STH    3,8(6)                    NN
          M      2,=F'100'
          SR     4,3
          STH    4,10(6)                   BB
F40       LA     6,12(6)
          BCT    10,F15
          RETURN (2,10)
F45       RETURN (2,10),RC=4
WD        DC     F'0'                      STANDARD WORD BOUNDRY
*                                          BCD TO EBCDIC TABLE
TABLE     DC     XL16'F0F1F2F3F4F5F6F7F8F9404040404040'
          DC     XL16'4040E2E3E4E5E6E7E8E9404040404040'
          DC     XL16'40D1D2D3D4D5D6D7D8D9404040404040'
          DC     XL16'40C1C2C3C4C5C6C7C8C9404040404040'
          LTORG
          END
```

APPENDIX D

SUBROUTINE GETS PROGRAM LISTING

GETS — DATA EXTRACTOR

THIS SUBROUTINE IS DESIGNED TO USE DATA BLOCK FORMAT DESCRIPTION TO ACQUIRE DATA BY A FORTRAN USER. IT IS USED AS A FUNCTION SUBROUTINE.

ONCE THE DATA, INCLUDING THE FORMAT TABLES, ARE IN CORE, PARTICULAR DATA ITEMS MAY BE ACQUIRED BY MEANS OF THE ENTRIES, GET AND IGET. EACH OF THESE ENTRIES WILL LOCATE ANY DATA ITEM, UNPACK IT, AND INTERPRET IT ACCORDING TO THE INFORMATION IN THE FORMAT TABLE. THE ITEM IS RETURNED AS A BINARY INTEGER (IN THE CASE OF IGET) OR AS A FLOATING POINT NUMBER (IN THE CASE OF GET).

```
* CALLING SEQUENCES
*
* GET (OR IGET) REQUIRES THREE ARGUMENTS:
*
* GET(FORMAT,BASE,ITEM)
*
* WHERE  FORMAT IS THE RELEVANT FORMAT TABLE ADDRESS.
*        BASE   IS THE BASE ADDRESS OF THE DATA BLOCK DESIRED.
*        ITEM   IS THE NUMBER OF THE SPECIFIC ITEM DESIRED IN
*               THAT DATA BLOCK.
*
*        VERSION 04/17/70
*
         START 0
         ENTRY IGET                      FIX POINT OUTPUT
         ENTRY GET                       FLOATING POINT OUTPUT
         USING *,15
IGET     B     G5
         DC    XL1'05',CL5'IGET'         TRACE NAME
G5       MVI   KEY,X'01'                 IGET KEY
         B     G15
         USING *,15
GET      B     G10
         DC    XL1'03',CL3'GET'          TRACE NAME
G10      MVI   KEY,X'02'                 GET KEY
G15      STM   2,11,28(13)               SAVE FORTRAN REGISTERS USED
         BALR  11,0
         USING *,11
         LM    7,9,0(1)
*                                        7  FORMAT ADR
*                                        8  BASE NUMBER ADDRESS
*                                        9  ITEM NUMBER ADDRESS
*                                        10 M
         L     7,0(7)
         L     8,0(8)                    BASE ADDRESS
         L     5,0(9)                    ITEM NUMBER
         S     5,=F'1'
         M     4,=F'12'                  CONVERT TO BYTES
         AR    7,5                       ITEM ADDRESS
         SR    2,2
         LH    10,2(7)
         CR    2,10                      TEST M
         BNE   G35
```

```
         MVC   WD(4),8(7)          FLOATING POINT WORD
         B     G122
G20      TM    KEY,X'02'           TEST FOR FLOATING ANSWER
         BC    G30
G25      RETURN (2,11)
G30      O     0,=X'46000000'      EXPONENT
         ST    0,WD
G32      SER   0,0                 NORMALIZE
         AE    0,WD                GET
         B     G25
G35      LH    3,4(7)              RRRR
         S     3,=F'1'
         CH    2,0(7)              TEST P
         BNE   G40
         LR    1,3                 MULTIPLY RRRR BY 3
         AR    3,1
         AR    3,1
G40      AR    8,3                 DATA ADDRESS
         C     10,=F'6'            TEST M
         BE    G135                224 FLOATING POINT
         BH    G120                360 FLOATING POINT
*                                  1 DATA
*                                  2 ZERO
*                                  4 XX
*                                  5 NN
*                                  6 BB
*                                  7 M
         LH    4,6(7)
         CR    4,2                 TEST XX
         BNE   G45
         LA    4,1                 XX=1
         LA    5,24                NN=24
         LA    6,24                BB=24
         B     G50
G45      LH    5,8(7)              NN
         LH    6,10(7)             BB
*                                  LEFT ADJUST DATA
G50      SR    9,9                 POSITION NUMBER
         MVC   WD(4),0(8)           DATA ON WORD BOUNDRY
         L     3,WD                LOAD DATA
         S     4,=F'1'
         SLL   3,0(4)              LEFT ADJUST DATA
         AR    10,10               MULTIPLY M BY 4
         AR    10,10
         B     GG-4(10)
GG       B     G55                 M=1 POSITIVE
         B     G60                 M=2 SIGN MAGNITUDE
         B     G60                 M=3 SIGN MAGNITUDE
         B     G70                 M=4 2'S COMPLEMENT
         B     G80                 M=5 1'S COMPLEMENT
G55      SLDL  2,0(5)              POSITIVE DATA
         B     G85
G60      CR    3,2                 TEST DATA
         BNL   G55                 POSITIVE
         SLL   3,1                 DROP SIGN BIT
```

```
        SRL     3,1
G65     LA      9,1             NEGATIVE NUMBER
        B       G55
G70     CR      3,2             2'S COMPLEMENT
        BNL     G55             POSITIVE
G75     SR      2,3             MAKE POSITIVE
        LR      3,2
        SR      2,2
        B       G65
G80     CR      3,2             1'S COMPLEMENT
        BNL     G55             POSITIVE
        SLDL    2,0(5)
        LA      2,1(2)          CONVERT TO 2'S COMPLEMENT
        SRDL    2,0(5)
        B       G75
G85     TM      KEY,X'02'       TEST ENTRY
        BO      G100            FLOAT
        CR      5,6             TEST NN,BB
        BH      G90             NN>BB
        SR      6,5             BB-NN
        SLL     2,0(6)
        B       G95             EXIT
G90     SR      5,6             NN-BB
        SRL     2,0(5)
G95     LR      0,2             ANSWER
        SR      1,1
        CR      9,1             TEST SIGN
        BE      G25             POSITIVE
        SR      1,2             MAKE NEGATIVE
        LR      0,1
        B       G25
G100    CR      6,5             TEST BB,NN
        BL      G110            BB<NN
        SR      6,5             BB-NN
        LR      1,6
        N       1,=F'3'
        SLL     2,0(1)          POWER OF 2
        SRL     2,2             POWER OF 16
        LA      6,70(6)         EXPONENT
G105    SLL     6,24            POSITIONAL EXPONENT
        SLL     9,31            SIGN BIT
        OR      6,9             SIGN+EXPONENT
        OR      2,6             ANSWER
        ST      2,WD
        LE      0,WD
        B       G25
G110    SR      5,6             NN-BB
        LR      1,5
        SRL     5,2             POWER OF 16
        N       1,=F'3'         POWER OF 2
        SR      3,3
        CR      1,3             ZERO POWER OF 2
        BE      G115
        LA      5,1(5)
        LA      3,4
```

```
         SR     3,1
         SLL    2,0(3)
G115     LA     6,70                    STANDARD EXPONENT
         SR     6,5
         B      G105
G120     MVC    WD(4),0(8)              360 FLOAT DATA
G122     TM     KEY,X'02'
         BO     G32
         L      3,WD                    LOAD DATA
         SR     2,2
         SLDL   2,1                     SIGN BIT
         LR     9,2
         SR     2,2
         SLDL   2,7                     EXPONENT
         SRL    3,8
         LA     1,70
         CR     2,1
         BL     G130                    E<70
         SR     2,1
         SLL    2,2                     MULTIPLY BY 2
         SLL    3,0(2)
G125     LR     0,3
         SR     2,2
         CR     2,9
         BE     G25
         SR     2,3                     MAKE NEGATIVE
         LR     0,2
         B      G25
G130     SR     1,2                     DROP BITS
         SLL    1,2
         SRL    3,0(1)
         B      G125
G135     MVI    WD+7,X'00'
         MVI    WD,X'00'
         LA     2,0
         MVC    WD+1(3),0(8)
         LA     3,0
         MVC    WD+4(3),3(8)
         L      4,WD+4
TESTMAN  LTR    5,4
         SLL    4,1
         BC     10,TSTEXP
         MVI    MANFLG,X'80'
TSTEXP   TM     WD+2,X'01'
         BZ       POSEXP
         MVI    EXPFLG,X'FF'
         XI     WD+3,X'FF'
         IC     3,WD+3
         A      3,=F'1'
         D      2,=F'4'
         STC    2,SHIFT+3
         BC      15,SHIFT
POSEXP   IC     3,WD+3
         D      2,=F'4'
         C      2,=F'0'
```

```
            BC      8,REMZERO
            LA      5,4
            SR      5,2
            STC     5,SHIFT+3
SHIFT       SRL     4,0
            BC      15,NEWEXP
REMZERO     S       3,=F'1'
NEWEXP      LA      5,65
            TM      EXPFLG,X'FF'
            BC      1,NEGEXP
            AR      5,3
            BC      15,STFLPT
NEGEXP      SR      5,3
            S       5,=F'1'
STFLPT      SRL     4,8
            ST      4,WD
            STC     5,WD
            OC      WD(1),MANFLG
            B       G122
EXPFIG      DC      X'00'
MANFIG      DC      X'00'
WD          DS      2F
KEY         DS      CL1
            LTORG
            LTORG
            END
```